FYS4260-2011

Kretssimulering
med CadStar og PSpice.

" *how to* ? "



28.02.2011
S. Lyng Nielsen
ELAB
Dept. of Physics
University of Oslo.

Design prosessen:

- CADSTAR settes opp iht. kursbeskrivelsen for FYS4260.
- Skjema tegnes
- Export to File : velg format : PSpice.
  (OBS på Current Sheet versus Whole Design)
  OPTIONS : "Use Header" (velg file xxxx.IN)
- Transfer to PCB : utlegget plasseres & Routes



CADSTAR PSpice Format :

AnOverallView
FormatofaPSPICEoutputfile
HowtogenerateaPSPICEoutput

Description

Selection of this format converts a Schematic design into the format recognised by the PSpice program.
For example:

C5 12 0 CMOD 10PF

When the PSpice format is selected, the following parameters on the dialog are also active:

Options:

This parameter display a dialog which enables you to read in a Header file, containing PSpice commands:

Source Data:

This parameter enables you to specify which sheets in the Schematic are exported in the PSpice format. The Select Sheets button displays a dialog on which you can select individual sheets for export.

Before Exporting PSpice ...

You must ensure that the following rules are obeyed:

Symbols

They must have been allocated to Parts in the Parts library. In addition, these Part entries must contain the appropriate ***SPI** line.

MoreInformationOnThePartsLibraryFormat...

Symbol Names

They must use a naming convention which satisfies the PSpice device name format. This format requires the first character of a component name to identify a PSpice device. For example, a capacitor name must always start with a 'C'.

TheListOfFirstCharactersRequiredForPSpiceNames...

PSPICE Commands

PSpice allows you to input additional PSpice **commands**. These commands set up circuit parameters model descriptions, types of simulation, etc. You need to type these commands in from the keyboard, and save them in a file with the extension **.in**

PSpice will merge the generated circuit description with these PSPICE commands.

Note: When using a PSpice commands file, you need to insert the converted CADSTAR circuit description within it. To specify the point of insertion you use the following command:

.CADSPICE

The circuit description will be inserted at the location of .CADSPICE

Refer to your PSpice manual for more information on these commands.

OPPGAVE-1:

Følgende krets tegnes og simuleres: (Passivt lav & høy – pass filter)



Eksempel på PSPICE–nettliste:

```
* Project FYS4260_2011-02-04 ( NOT FROM CADSTAR )

R1     INP              OUT_LP  R01%          10.0K
C1     INP              OUT_HP C10%           10NF
R2     OUT_HP    0              R01%          10.0K
C2     OUT_LP    0              C10%          10NF

VIN    INP       0  AC 1 0 SIN( 0V 1V 1KHZ 0 0 0 )

.TRAN  1US 2MS
.AC  DEC 10 1HZ 100MEGHZ

.MODEL R01%     RES(R=1 DEV=1% TC1=.00005 TC2=.00001)
.MODEL C10%     CAP(C=1 DEV=10% TC1=0.0004);400ppm

.PROBE
.END
```

Cadstar eksporterer dessverre svært ”primitiv” Pspice nettliste.
Tar ikke med NODENAVN.
Skiller ikke mellom simulerbare komponenter og ANNET.
(brytere, kantkontakter – osv)

Det overstående skjema vil bli eksportert som:

C2	1	0	C10%	10NF
R3	2	0	R01%	10K
C1	2	3	C10%	10NF
R1	1	3	R01%	10K

Ganske så mangelfullt som praktisk input , mao…

LITT hjelpes på med å ha en **CadStar.in** file - en **HEADER** fil...
(i File export, OPTIONS: USE HEADER.)

En Header file kan inneholde nødvendig tilleggsinformasjon en gyldig Pspice file trenger for automatisert simuleringsprosess.

EKSEMPEL på CADSTAR.IN file :

```
* CADSTAR_PSPICE CONTROLL FILE
*-----------------------
* INSERT NODES FOR XX's

*GENERATORS:
Vin  XX  0  AC 1
*VSin XX  0  SIN ( 0V 0.1V 1KHZ 0 0 0 )

*POWER SUPPLY
VDC1  XX   XX  15V
VDC2  XX   XX -15V
*-----------------------
.CADSPICE
*-----------------------
*TESTS:
.AC OCT 10 10 100MEG
*.TRAN  1US 2MS

*MODELS
.MODEL C10%      CAP(C=1 DEV=20% TC1=0.00015);150ppm
.MODEL R01%      RES(R=1 DEV=1% TC1=.00005 TC2=.00001)

.LIB
.PROBE
.END
```

NÅR det benyttes en HEADER fil, vil Cadstar Pspice filen bli flettet inn i HEADER filen der statementet : **. CADSPICE** står

Eksempel:

```
* SPICE CONTROLL FILE
*----------------------
* INSERT NODES FOR XX's

*GENERATORS:
Vin X  0  SIN ( 0V 0.1V 1KHZ 0 0 0 )

*POWER SUPPLY
*VDC1  XX   XX  15V
*VDC2  XX   XX -15V

C2      1       0       C10%            10NF
R3      2       0       R01%            10K
C1      2       3       C10%            10NF
R1      1       3       R01%            10K

*TESTS:
.AC OCT 10 10 100MEG
*.TRAN  1US 2MS

*MODELS
.MODEL C10%     CAP(C=1 DEV=20% TC1=0.00015);150ppm
.MODEL R01%     RES(R=1 DEV=1% TC1=.00005 TC2=.00001)

.LIB
.PROBE
.END
```

Editeres inn rette nodenumre for INP , OUT_LP, OUT_HP – osv. er filen klar for Pspice simulering.

```
* SPICE CONTROLL FILE
*----------------------
* INSERT NODES FOR XX's

*GENERATORS:
Vin  3  0  AC 1 SIN ( 0V 0.1V 1KHZ 0 0 0 )

*POWER SUPPLY
*VDC1  XX   XX  15V
*VDC2  XX   XX -15V

C2 1 0  C10% 10NF
R3 2 0  R01% 10K
C1 2 3  C10% 10NF
R1 1 3  R01% 10K

*TESTS:
.AC OCT 10 10 100MEG
*.TRAN  1US 2MS

*MODELS
.MODEL C10%      CAP(C=1 DEV=20% TC1=0.00015);150ppm
.MODEL R01%      RES(R=1 DEV=1% TC1=.00005 TC2=.00001)

.LIB
.PROBE
.END
```

Pspice viser følgende resultat av node 1 & 2 i *dB* :



EKSEMPEL på litt mer ”utviklet” CADSTAR.IN file :

```
* SPICE CONTROLL FILE
*-----------------------

* INSERT NODES FOR XX's

*GENERATORS :

Vin  XX  0  AC 1 SIN ( 0V 0.1V 1KHZ 0 0 0 )

*POWER SUPPLY

VDC1  XX   XX  9V
VDC2  XX   XX -9V

*-------------------------
.CADSPICE
*-------------------------

*DUMMY TEST COMPONENTS:
Rload 2   0   10R
*-------------------------

*TESTS :

.AC OCT 10 10 100MEG
*.NOISE V(2) VIN 100

*.MC 10 AC VDB(2) YMAX LIST
*.WCASE AC VDB(2) YMAX LIST

*.TRAN  1US 2MS
*.FOUR 1KHZ V(2)

.DC  VIN -0.125V 0.125V 0.005V

*.MODEL CMOD CAP(C=10PF)
*.STEP CAP CMOD(C) LIST 1P 10PF 100PF

*.MODEL RMOD RES(R=1K)
*.STEP RES RMOD(R) LIST 1K 2K 3K

*-------------------------
```

```
* MODELS :
*--------------------------------

* NPN SMALL SIGNAL THROUGH HOLE :

.MODEL BC550C NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=23.89 Bf=493.2 Ise=99.2f
+Ne=1.829 Ikf=.1542 Nk=.6339 Xtb=1.5 Br=2.886 Isc=7.371p
+Nc=1.508 Ikr=5.426 Rc=1.175 Cjc=5.5p Mjc=.3132 Vjc=.4924 Fc=.5
+Cje=11.5p Mje=.6558 Vje=.5 Tr=10n Tf=420.3p Itf=1.374 Xtf=39.42
+Vtf=10)
*PHILIPS                    PID=BC550C        CASE=TO92

* NPN SMALL SIGNAL SMD :

.MODEL BC817 NPN IS = 9.198E-14 NF = 1.003 ISE = 4.468E-16 NE = 1.65
+BF = 338.8  IKF = 0.4913 VAF = 107.9 NR = 1.002
+ISC = 5.109E-15 NC = 1.071  BR = 29.48 IKR = 0.193
+VAR = 25 RB = 1 IRB = 1000 RBM = 1  RE = 0.2126 RC = 0.143
+XTB = 0 EG = 1.11  XTI = 3 CJE = 3.825E-11
+VJE = 0.7004 MJE = 0.364 TF = 5.229E-10 XTF = 219.7
+VTF = 3.502  ITF = 7.257 PTF = 0 CJC = 1.27E-11 VJC = 0.4431
+MJC = 0.3983  XCJC = 0.4555 TR = 7E-11
+CJS = 0 VJS = 0.75 MJS = 0.333 FC = 0.905

.MODEL BC817-40 NPN(Is=32.53f Xti=3 Eg=1.11 Vaf=100 Bf=407.8 Ise=333.4f
+Ne=1.692 Ikf=.9448 Nk=.6041 Xtb=1.5 Br=6.39 Isc=268.8f Nc=2.125
+Ikr=.1193 Rc=.4205 Cjc=24.29p Mjc=.3333 Vjc=.5 Fc=.5 Cje=71.14p
+Mje=.3333 Vje=.5 Tr=10n Tf=662p Itf=3.238 Xtf=134.9 Vtf=10)
*                    SIEMENS                    pid=bc817-40       case=SOT23

.MODEL BC850C   NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=62.37 Bf=279.1 Ise=59.16f
+Ne=1.81 Ikf=.2201 Nk=.6305 Xtb=1.5 Br=3.816 Isc=16.17f Nc=2.394
+Ikr=1.859 Rc=1.508 Cjc=7.287p Mjc=.3333 Vjc=.5 Fc=.5 Cje=9.485p
+Mje=.3333 Vje=.5 Tr=10n Tf=664p Itf=4.664 Xtf=147 Vtf=10)
*SIEMENS                    PID=BC850C        CASE=SOT23

.MODEL FMBM5551 NPN IS= 5.88844E-15     BF= 76.1  NF     = 1
+BR     = 0.028          NR     = 1            ISE    = 1.0E-14
+NE     = 1.56           ISC    = 1.01826E-14    NC     = 1.5
+VAF    = 74.03          VAR    = 20           IKF    = 0.049453
+IKR    = 0.017378       RB     = 12.5         RBM    = 1.2
+IRB    = 2.51189E-5     RE     = 0.11         RC     = 1.03
+CJE    = 2.079424E-12    VJE    = 0.6464066     MJE    = 0.2596434
+FC     = 0.5            CJC    = 1.091838E-12   VJC    = 0.5
+MJC    = 0.1964945      TF     = 1.012E-09     XTF    = 2.0
+ITF    = 0.4            VTF    = 4.0          TR     = 2.395E-7
+XTB    = 1.65           EG     = 0.82
+XTI    = 3
*FAIRCHILD       PID=FMBM5551   CASE=SUPERSOT-6

.MODEL MMBT5551 NPN (Is=2.511f Xti=3 Eg=1.11 Vaf=100 Bf=242.6
+Ne=1.249 Ise=2.511f Ikf=.3458 Xtb=1.5 Br=3.197
+Nc=2 Isc=0 Ikr=0 Rc=1 Cjc=4.883p Mjc=.3047 Vjc=.75
+Fc=.5 Cje=18.79p Mje=.3416 Vje=.75 Tr=1.202n Tf=560p
+Itf=50m Vtf=5 Xtf=8 Rb=10)
*FAIRCHILD       PID=MMBT5551   CASE=SOT-23
```

```
* PNP SMALL SIGNAL THROUGH HOLE :
*---------------------------------
.MODEL BC560C   PNP(Is=1.02f Xti=3 Eg=1.11 Vaf=34.62 Bf=401.6 Ise=38.26p
+Ne=5.635 Ikf=74.73m Nk=.512 Xtb=1.5 Br=9.011 Isc=1.517f
+Nc=1.831 Ikr=.1469 Rc=1.151 Cjc=9.81p Mjc=.332 Vjc=.4865 Fc=.5
+Cje=30p Mje=.3333 Vje=.5 Tr=10n Tf=524p Itf=.9847 Xtf=17.71
+Vtf=10)
*PHILIPS                    PID=BC560C        CASE=TO92

* PNP SMALL SIGNAL SMD :
*---------------------------------
.MODEL BC807 PNP IS=1.08E-13 NF=0.99 ISE=2.713E-14
+NE=1.4 BF=385.7 KF=0.3603 VAF=31.29 NR=0.9849 ISC=5.062E-13
+NC=1.295 BR=20.57 IKR=0.054 VAR=11.62 RB=1 IRB=1E-06
+RBM=0.5 RE=0.1415 RC=0.2623 XTB=0 EG=1.11 XTI=3
+CJE=5.114E-11 VJE=0.8911 MJE=0.4417 TF=7.359E-10 XTF=1.859
+VTF=3.813 ITF=0.4393 PTF=0 CJC=2.656E-11 VJC=0.62 MJC=0.4836
+XCJC=0.459 TR=5.00E-08 CJS=0 VJS=0.75 MJS=0.333 FC=0.99

.MODEL BC807-40 PNP(Is=32.53f Xti=3 Eg=1.11 Vaf=100 Bf=460.8 Ise=301.2f
+Ne=1.615 Ikf=.4967 Nk=.5594 Xtb=1.5 Br=6.87 Isc=283.2f Nc=1.97
+Ikr=9.135m Rc=.3759 Cjc=24.29p Mjc=.3333 Vjc=.5 Fc=.5
+Cje=71.14p Mje=.3333 Vje=.5 Tr=10n Tf=704.3p Itf=13.89
+Xtf=1.333K Vtf=10)
*SIEMENS                    PID=BC807-40      CASE=SOT23

.MODEL FMBM5401 PNP IS= 9.0095E-14  BF= 91.0  NF = 1
+ BR= 0.2745 NR= 1  ISE= 1.059573E-12
+ NE     = 2           ISC    = 1.5578E-09     NC     = 1.5
+ VAF    = 86.87       VAR    = 39.22          IKF    = 0.9394
+ IKR    = 0.831764    RB     = 18.4           RBM    = 0.0126
+ IRB    = 1.0256E-7   RE     = 0.048          RC     = 1.5212
+ CJE    = 2.61493E-12 VJE    = 0.7535019      MJE    = 0.3860237
+ FC     = 0.5         CJC    = 1.882005E-11   VJC    = 0.5964168
+ MJC    = 0.3948966   TF     = 1.57155e-09    EG     = 0.7005
+ XTB    = 1.115
*FAIRCHILD       PID=FMBM5401   CASE=SUPERSOT-6


*DRIVERS:
*---------------------------------

* NPN  THROUGH HOLE :

.MODEL Q2SC3423 NPN IS=9.98627F BF=2K NF=967.67M VAF=100 IKF=49.6929M ISE=1.04163F
+ NE=1.07574 BR=601.257M IKR=462.798U ISC=32.904P RC=899.97M
+ CJE=2P MJE=500M CJC=6.42174P VJC=749.999M MJC=499.509M
+ TF=713.346P XTF=500M VTF=10 ITF=9.9976M TR=10N
*TOSHIBA                    PID=Q2SC3423     CASE=2-8H1A

* NPN  SMD :

.MODEL FZT651 NPN IS =2.218E-13 NF =.9956 BF =230 IKF=2 VAF=100 ISE=2.9E-14
+NE =1.35 NR =.995 BR =56 IKR=1 VAR=30 ISC=2.971E-13 NC =1.321 RB =.04
+RE =.075 RC =.069 CJC=51E-12 MJC=.42 VJC=.595 CJE=318E-12 TF =.77E-9
+TR =27E-9
*ZETEX  PID=FZT651  CASE=SOT-223
```

* PNP THROUGH HOLE :

.MODEL Q2SA1360 PNP IS=10F BF=134.853 VAF=100 IKF=109.96M ISE=221.874F
+ NE=1.66575 BR=10 IKR=880.176M ISC=187.58P NC=1.90472 RE=1 RC=15.5104
+ CJE=2P MJE=500M CJC=6.24728P VJC=692.028M MJC=340.013M TF=1.08385N
+ XTF=16.9293 VTF=9.36211 ITF=670.025M TR=10N
*TOSHIBA PID=Q2SA1360 CASE=2-8H1A

* PNP SMD :

.MODEL FZT751 PNP IS =2.715E-13 BF =170 VAF=70 NF =1.004 IKF=2.75 ISE=1E-13
+NE =1.535 BR =23 VAR=40 NR =1.005 IKR=.55 ISC=5.15E-14 NC =1.13 RB =.07
+RE =.065 RC =.085 CJE=360E-12 TF =.94E-9 CJC=90E-12 TR =60E-9 VJC=.705
+MJC=.46
*ZETEX PID=FZT751 CASE=SOT-223

* DIODES :
*---

* THROUGH HOLE :

.MODEL D1N4148 D(Is=2.682n N=1.836 Rs=.5664 Ikf=44.17m Xti=3 Eg=1.11 Cjo=4p
+M=.3333 Vj=.5 Fc=.5 Isr=1.565n Nr=2 Bv=100 Ibv=100u Tt=11.54n)
*TELEFUNKEN PID=D1N4148 CASE=DO35

* SMD :
.MODEL BAS16 D(Is=4.198n N=1.977 Rs=.4599 Ikf=1.408 Xti=3 Eg=1.11 Cjo=2p
+M=.3333 Vj=.5 Fc=.5 Isr=1.115n Nr=2 Bv=75 Ibv=100u Tt=8.656n)
*SIEMENS PID=BAS16 CASE=SOT23

.MODEL BAV70 D(Is=1.211n N=1.797 Rs=.8835 Ikf=0 Xti=3 Eg=1.11
+ Cjo=1.5p M=.3333 Vj=.5 Fc=.5 Isr=1.689n Nr=2 Tt=8.656n)
*SIEMENS PID=BAV70 CASE=SOT23

* DIV. MODELS :
*---

.MODEL C025% CAP(C=1 DEV=2.5% TC1=0.00015);150ppm
.MODEL C5% CAP(C=1 DEV=5% TC1=0.00015);150ppm
.MODEL C20% CAP(C=1 DEV=20% TC1=0.00015);150ppm

.MODEL R01% RES(R=1 DEV=1% TC1=.00005 TC2=.00001)

.OP
.LIB
.PROBE/CSDF
.END

HVA forventes av Pspice simuleringen, og hva skal rapporteres i FYS4260 ?

(Simuleringsrapporten leveres inn sammen med hovedrapporten)

Noder identifiseres og spenninger fra simuleringen (.OUT – fila) noteres.



Inspiser resultat fra BodePlot & Osc. analysene :



PRIMÆRT:

Virker kretsen som ønsket? – Evnt. Hvorfor ikke ?

Betrakt DC node spenningene og strømtrekk + effektforbruk fra batteriene.

Fra Bodeplottet (Gain & Loss – versus frequency)
Identifiser GAIN@ 1Khz. og -3dB punktene.

Fra Oscilloscop : (Voltage versus time)
Identifiser OFFSET spenning og Peak to Peak Voltage sving.

Forslag til forbedringer / endringer ??